STORED PROCEDURES AND FUNCTIONS

PL/SQL programs can be of two types:  procedures and functions.  They help in devloping applications that provide extensibility, modularity.  These programs are reusable and hence help in maintaining an application development environment. A stored procedure or function is a PL/SQL program unit that 

· has a specific name 

· can take parameters  as input, and return values as output

Once written and compiled, the procedure can be invoked repeatedly by many users. The details of a procedure are stored in the data dictionary. So, individual users need not write the query time. They also can be refined and improved after repeated use resulting in an improved query available for everyone’s use. Functions are similar to stored procedures.  They perform a specific task within an application and are generally used to modularize an application. 

The general structure of a stored procedure is:

PROCEDURE procedurename [(param1, param2,…)] IS

[declaration section]

BEGIN


Statements

[EXCEPTION

exception handlers]

END [procedure name];
Once written and compiled, a PL/SQL procedure must be called explicitly to execute it. Unlike triggers they are NOT automatically executed. Syntax for calling a stored procedure is:

Procedurename [(param1, param2,…)]





The formal parameters of a procedure have three major parts: 

 name - The name of the parameter, which must be a legal PL/SQL identifier.  

 mode  -The parameter mode, which indicates whether the parameter is an input-only parameter (IN), an output-only parameter (OUT), or is both an input and an output parameter (IN OUT). If the mode is not specified, IN is assumed.  

 datatype  - The parameter datatype is a standard PL/SQL datatype.

Parameters for a procedure can be used in three modes:

IN
allows values to be passed into a procedure. These values can be subsequently used within the procedure but cannot be assigned a value within the procedure. IN is the default mode.  This means that if a mode is not specified for a parameter it will be treated as an input parameter.

OUT
allows values computed within the procedure to be returned to the calling program.

INOUT
allows for passing initial values to a procedure and the updated values are then returned to the calling program.

The three parameter modes, IN (the default), OUT, and IN OUT, can be used with any subprogram. However, it is preferable to avoid using the OUT and IN OUT modes with functions. The purpose of a function is to take any arguments, if needed, to return a single value. It is poor programming practice to have a function return multiple values. Also, functions should be free from side effects, which change the values of variables not local to the subprogram.

	IN
	OUT
	IN OUT

	the default
	must be specified
	must be specified

	passes values to a subprogram
	returns values to the caller
	passes initial values to a subprogram; returns updated values to the caller

	formal parameter acts like a constant
	formal parameter acts like an uninitialized variable
	formal parameter acts like an initialized variable

	formal parameter cannot be assigned a value
	formal parameter cannot be used in an expression; must be assigned a value
	formal parameter should be assigned a value

	actual parameter can be a constant, initialized variable, literal, or expression
	actual parameter must be a variable
	actual parameter must be a variable


Consider the following example of a procedure:

PROCEDURE get_partnames (cost_cent IN number IS

    item_name       VARCHAR2(10);

    CURSOR         mycursor (cost_center NUMBER) IS

                      SELECT item_name FROM item_

                        WHERE cost_center = cost_cent;

BEGIN

    OPEN mycursor(cost_cent);

    LOOP

        FETCH mycursor INTO item_name;

        EXIT WHEN mycursor %NOTFOUND;

        DBMS_OUTPUT.PUT_LINE(item_name);

    END LOOP;

    CLOSE mycursor;

END;

Parameter Datatypes 

The datatype of a formal parameter consists of one of the following: 

· an unconstrained type name, such as NUMBER or VARCHAR2. Numerically constrained types such as NUMBER(2) or VARCHAR2(20) cannot be used in a parameter list.

· a type that is constrained using the %TYPE or %ROWTYPE attributes 

%TYPE and %ROWTYPE Attributes 

However, you can use the type attributes %TYPE and %ROWTYPE to constrain the parameter. For example, the GET_EMP_NAMES procedure specification in "Procedure Parameters" could be written as 

PROCEDURE get_partnames (cost_cent IN Item.cost_center%TYPE)
to have the cost_cent parameter take the same datatype as the cost_center column in the Item table. The column and table must be available when a declaration using %TYPE (or %ROWTYPE) is used.  As always, using %TYPE is preferable, since if the type of the column in the table changes, it is not necessary to change the application code. 

If the get_partnames procedure is part of a package, then you can use previously-declared public (package) variables to constrain a parameter datatype. For example: 

cost_cent    number(2);

...

PROCEDURE get_emp_names(cost_ce IN cost_cent %TYPE);
You use the %ROWTYPE attribute to create a record that contains all the columns of the specified table. The following example defines the get_part_rec procedure, which returns all the columns of the EMP table in a PL/SQL record, for the given EMPNO: 

PROCEDURE get_part_rec (cost_cent  IN Item.cost_center%TYPE,

                       item_ret    OUT item%ROWTYPE) IS

BEGIN

    SELECT item_id, item_name, item_price, cost_center

        INTO item_ret

        FROM item

        WHERE cost_center = cost_cent;

END;

You could call this procedure from a PL/SQL block as follows:  (for now assume that there is only one row for this cost_center)

DECLARE

  item_row      Item%ROWTYPE;     -- declare a record matching a row in the item table

BEGIN

  get_part_rec(7499, emp_row);   -- call for emp# 7499

  DBMS_OUTPUT.PUT(emp_row.ename || ' ' || emp_row.empno);

  DBMS_OUTPUT.PUT(' ' || emp_row.job || ' ' || emp_row.mgr);

  DBMS_OUTPUT.PUT(' ' || emp_row.hiredate || ' ' || emp_row.sal);

  DBMS_OUTPUT.PUT(' ' || emp_row.comm || ' ' || emp_row.deptno);

  DBMS_OUTPUT.NEW_LINE;

END;

Stored functions can also return values that are declared using %ROWTYPE. For example: 

FUNCTION get_emp_rec (dept_num IN emp.deptno%TYPE) RETURN emp%ROWTYPE IS ...

Tables and Records 

You can pass PL/SQL tables as parameters to stored procedures and functions. You can also pass tables of records as parameters. 

Default Parameter Values 

Parameters can take default values. You use the DEFAULT keyword or the assignment operator to give a parameter a default value. For example, the specification for the GET_EMP_NAMES procedure  could be written as 

PROCEDURE get_emp_names (dept_num IN NUMBER DEFAULT 20) IS ...

or as 

PROCEDURE get_emp_names (dept_num IN NUMBER := 20) IS ...

When a parameter takes a default value, it can be omitted from the actual parameter list when you call the procedure. When you do specify the parameter value on the call, it overrides the default value. 

DECLARE Keyword 

Unlike in an anonymous PL/SQL block, you do not use the keyword DECLARE before the declarations of variables, cursors, and exceptions in a stored procedure. In fact, it is an error to use it. 

Creating Stored Procedures and Functions 

Use your normal text editor to write the procedure. At the beginning of the procedure, place the command 

CREATE PROCEDURE procedure_name AS   ...

For example, to use the example on page 10-8, you can create a text (source) file called get_emp.sql containing the following code: 

CREATE PROCEDURE get_emp_rec (emp_number  IN emp.empno%TYPE,

                              emp_ret    OUT emp%ROWTYPE) AS

BEGIN

    SELECT empno, ename, job, mgr, hiredate, sal, comm, deptno

        INTO emp_ret

        FROM emp

        WHERE empno = emp_number;

END;

/

Then, using an interactive tool such as SQL*Plus, load the text file containing the procedure by entering the command 

SQLPLUS> @get_emp

to load the procedure into the current schema from the get_emp.sql file (.sql is the default file extension). Note the slash (/) at the end of the code. This is not part of the code; it just activates the loading of the procedure. 

 WARNING: 

When developing a new procedure, it is usually much more convenient to use the CREATE OR REPLACE... PROCEDURE command. This replaces any previous version of that procedure in the same schema with the newer version, but note that this is done without warning. 

You can use either the keyword IS or AS after the procedure parameter list. 

Use the CREATE [OR REPLACE] FUNCTION... command to store functions. 

Privileges Required to Create Procedures and Functions 

To create a stand-alone procedure or function, or package specification or body, you must meet the following prerequisites: 

     You must have the CREATE PROCEDURE system privilege to create a procedure or package in your schema, or the

CREATE ANY PROCEDURE system privilege to create a procedure or package in another user's schema. 

            Note: 

            To create without errors, that is, to compile the procedure or package successfully, requires the following additional privileges: The owner of the procedure or package must have been explicitly granted the necessary object privileges for all objects referenced within the body of the code; the owner cannot have obtained required privileges through roles. 

If the privileges of a procedure's or package's owner change, the procedure must be reauthenticated before it is executed. If a necessary privilege to a referenced object is revoked from the owner of the procedure (or package), the procedure cannot be executed. 

The EXECUTE privilege on a procedure gives a user the right to execute a procedure owned by another user. Privileged users execute the procedure under the security domain of the procedure's owner. Therefore, users never have to be granted the privileges to the objects referenced by a procedure. This allows for more disciplined and efficient security strategies with database applications and their users. Furthermore, all procedures and packages are stored in the data dictionary (in the SYSTEM tablespace). No quota controls the amount of space available to a user who creates procedures and packages. 

Altering Stored Procedures and Functions 

To alter a stored procedure or stored function, you must first DROP it, using the DROP PROCEDURE or DROP FUNCTION command, then recreate it using the CREATE PROCEDURE or CREATE FUNCTION command. Alternatively, use the CREATE OR

REPLACE PROCEDURE or CREATE OR REPLACE FUNCTION command, which first drops the procedure or function if it exists, then recreates it as specified. 

          WARNING: 

          The procedure or function is dropped without any warning.

PL/SQL Packages 

A package is a group of PL/SQL types, objects, and stored procedures and functions. The specification part of a package declares the public types, variables, constants, and subprograms that are visible outside the immediate scope of the package. The body of a package defines the objects declared in the specification, as well as private objects that are not visible to applications outside the package. 

The following example shows a package specification for a package named EMPLOYEE_MANAGEMENT. The package contains one stored function and two stored procedures. 

CREATE PACKAGE employee_management AS

   FUNCTION hire_emp (name VARCHAR2, job VARCHAR2,

      mgr NUMBER, hiredate DATE, sal NUMBER, comm NUMBER,

      deptno NUMBER) RETURN NUMBER;

   PROCEDURE fire_emp (emp_id NUMBER);

   PROCEDURE sal_raise (emp_id NUMBER, sal_incr NUMBER);

END employee_management;

The body for this package defines the function and the procedures: 

CREATE PACKAGE BODY employee_management AS

   FUNCTION hire_emp (name VARCHAR2, job VARCHAR2,

      mgr NUMBER, hiredate DATE, sal NUMBER, comm NUMBER,

      deptno NUMBER) RETURN NUMBER IS

-- The function accepts all arguments for the fields in the employee table except for the employee number.

-- A value for this field is supplied by a sequence. The function returns the sequence number generated

-- by the call to this function.

       new_empno    NUMBER(10);

   BEGIN

      SELECT emp_sequence.NEXTVAL INTO new_empno FROM dual;

      INSERT INTO emp VALUES (new_empno, name, job, mgr,

         hiredate, sal, comm, deptno);

      RETURN (new_empno);

   END hire_emp;

   PROCEDURE fire_emp(emp_id IN NUMBER) AS

-- The procedure deletes the employee with an employee number that corresponds to the argument EMP_ID.  If  no employee is found, an exception is raised.

   BEGIN

      DELETE FROM emp WHERE empno = emp_id;

      IF SQL%NOTFOUND THEN

      raise_application_error(-20011, 'Invalid Employee

         Number: ' || TO_CHAR(emp_id));

   END IF;

END fire_emp;

PROCEDURE sal_raise (emp_id IN NUMBER, sal_incr IN NUMBER) AS

-- The procedure accepts two arguments.  EMP_ID is a

-- number that corresponds to an employee number.

-- SAL_INCR is the amount by which to increase the

-- employee's salary.

   BEGIN

-- If employee exists, update salary with increase.

      UPDATE emp

         SET sal = sal + sal_incr

         WHERE empno = emp_id;

      IF SQL%NOTFOUND THEN

         raise_application_error(-20011, 'Invalid Employee

            Number: ' || TO_CHAR(emp_id));

      END IF;

   END sal_raise;

END employee_management;

          Note: 

          If you want to try this example, first create the sequence number EMP_SEQUENCE. You can do this

          using the following SQL*Plus statement: 

          SQL> EXECUTE CREATE SEQUENCE emp_sequence

             > START WITH 8000 INCREMENT BY 10;

Creating Packages 

Each part of a package is created with a different command. Create the package specification using the CREATE PACKAGE command. The CREATE PACKAGE command declares public package objects. 

To create a package body, use the CREATE PACKAGE BODY command. The CREATE PACKAGE BODY command defines the procedural code of the public procedures and functions declared in the package specification. (You can also define private (or local) package procedures, functions, and variables within the package body.

The OR REPLACE Clause 

It is often more convenient to add the OR REPLACE clause in the CREATE PACKAGE or CREATE PACKAGE BODY commands when

you are first developing your application. The effect of this option is to drop the package or the package body without warning.

The CREATE commands would then be 

CREATE OR REPLACE PACKAGE package_name AS ...

and 

CREATE OR REPLACE PACKAGE BODY package_name AS ...

Privileges Required to Create Packages 

The privileges required to create a package specification or package body are the same as those required to create a

stand-alone procedure or function; 

Creating Packaged Objects 

The body of a package can contain 

     procedures declared in the package specification 

     functions declared in the package specification 

     definitions of cursors declared in the package specification 

     local procedures and functions, not declared in the package specification 

     local variables 

Procedures, functions, cursors, and variables that are declared in the package specification are global. They can be called, or used, by external users that have execute permission for the package, or that have EXECUTE ANY PROCEDURE privileges. 

When you create the package body, make sure that each procedure that you define in the body has the same parameters, by name, datatype, and mode, as the declaration in the package specification. For functions in the package body, the parametersas well as the return type must agree in name and type. 

Local Objects 

You can define local variables, procedures, and functions in a package body. These objects can only be accessed by other procedures and functions in the body of the same package. They are not visible to external users, regardless of the privileges they hold. 

Naming Packages and Package Objects 

The names of a package and all public objects in the package must be unique within a given schema. The package specification and its body must have the same name. All package constructs must have unique names within the scope of the package, unless overloading of procedure names is desired. 

Dropping Packages and Procedures 

A stand-alone procedure, a stand-alone function, a package body, or an entire package can be dropped using the SQL commands DROP PROCEDURE, DROP FUNCTION, DROP PACKAGE BODY, and DROP PACKAGE, respectively. A DROP PACKAGE statement drops both a package's specification and body. 

The following statement drops the OLD_SAL_RAISE procedure in your schema: 

DROP PROCEDURE old_sal_raise;

Privileges Required to Drop Procedures and Packages 

To drop a procedure or package, the procedure or package must be in your schema or you must have the DROP ANY

PROCEDURE privilege. An individual procedure within a package cannot be dropped; the containing package specification and body must be re-created without the procedures to be dropped. 

Package Invalidations and Session State 

Each session that references a package object has its own instance of the corresponding package, including persistent state for any public and private variables, cursors, and constants. If any of the session's instantiated packages (specification or body) are subsequently invalidated and recompiled, all other dependent package instantiations (including state) for the session are lost. 

For example, assume that session S instantiates packages P1 and P2, and that a procedure in package P1 calls a procedure in package P2. If P1 is invalidated and recompiled (for example, as the result of a DDL operation), the session S instantiations of both P1 and P2 are lost. In such situations, a session receives the following error the first time it attempts to use any object of an invalidated package instantiation: 

ORA-04068: existing state of packages has been discarded

The second time a session makes such a package call, the package is reinstantiated for the session without error. 

          Note: 

          Oracle has been optimized to not return this message to the session calling the package that it invalidated. Thus, in the example above, session S would receive this message the first time it called package P2, but would not receive it when calling P1. 

In most production environments, DDL operations that can cause invalidations are usually performed during inactive working hours; therefore, this situation might not be a problem for end-user applications. However, if package specification or body invalidations are common in your system during working hours, you might want to code your applications to detect for this error when package calls are made. For example, the user-side application might reinitialize any user-side state that depends on any session's package state (that was lost) and reissue the package call.

